

**symmetry** — Symmetry and marginal homogeneity tests

<a href="#">Description</a>	<a href="#">Quick start</a>	<a href="#">Menu</a>	<a href="#">Syntax</a>
<a href="#">Options</a>	<a href="#">Remarks and examples</a>	<a href="#">Stored results</a>	<a href="#">Methods and formulas</a>
<a href="#">References</a>	<a href="#">Also see</a>		

## Description

`symmetry` performs asymptotic symmetry and marginal homogeneity tests, as well as an exact symmetry test on  $K \times K$  tables where there is a 1-to-1 matching of cases and controls (nonindependence). This testing is used to analyze matched-pair case-control data with multiple discrete levels of the exposure (outcome) variable. In genetics, the test is known as the transmission/disequilibrium test (TDT) and is used to test the association between transmitted and nontransmitted parental marker alleles to an affected child (Spieldman, McGinnis, and Ewens 1993). For  $2 \times 2$  tables, the asymptotic test statistics reduce to the McNemar test statistic, and the exact symmetry test produces an exact McNemar test; see [R] [epitab](#). For many exposure variables, `symmetry` can optionally perform a test for linear trend in the log relative risk.

`symmetry` expects the data to be in the wide format; that is, each observation contains the matched case and control values in variables *casevar* and *controlvar*. Variables can be numeric or string.

`symmi` is the immediate form of `symmetry`. The `symmi` command uses the values specified on the command line; rows are separated by ‘\’, and options are the same as for `symmetry`. See [U] 19 [Immediate commands](#) for a general introduction to immediate commands.

## Quick start

Symmetry and marginal homogeneity tests for 1-to-1 matched case-control studies

```
symmetry case control
```

Same as above

```
symmetry control case
```

Exact test of table symmetry

```
symmetry case control, exact
```

Report the contribution from each off-diagonal pair to the overall  $\chi^2$ -statistic

```
symmetry control case, contrib
```

Test for a linear trend in the log of the relative risk

```
symmetry control case, trend
```

Request marginal homogeneity statistics that do not require the inversion of the variance-covariance matrix

```
symmetry case control, mh
```

Using frequency weight variable *wvar*

```
symmetry case control [fweight=wvar]
```

## Menu

### symmetry

Statistics > Epidemiology and related > Other > Symmetry and marginal homogeneity test

### symmi

Statistics > Epidemiology and related > Other > Symmetry and marginal homogeneity test calculator

## Syntax

*Symmetry and marginal homogeneity tests*

```
symmetry casevar controlvar [if] [in] [weight] [, options]
```

*Immediate form of symmetry and marginal homogeneity tests*

```
symmi #11 #12 [...] \ #21 #22 [...] [\...] [if] [in] [, options]
```

*options*

Description

Main

<code>notable</code>	suppress output of contingency table
<code>contrib</code>	report contribution of each off-diagonal cell pair
<code>exact</code>	perform exact test of table symmetry
<code>mh</code>	perform two marginal homogeneity tests
<code>trend</code>	perform a test for linear trend in the (log) relative risk (RR)
<code>cc</code>	use continuity correction when calculating test for linear trend

`fweights` are allowed; see [U] 11.1.6 [weight](#).

## Options

Main

`notable` suppresses the output of the contingency table. By default, `symmetry` displays the  $n \times n$  contingency table at the top of the output.

`contrib` reports the contribution of each off-diagonal cell pair to the overall symmetry  $\chi^2$ .

`exact` performs an exact test of table symmetry. This option is recommended for sparse tables.  
CAUTION: The exact test requires substantial amounts of time and memory for large tables.

`mh` performs two marginal homogeneity tests that do not require the inversion of the variance–covariance matrix.

By default, `symmetry` produces the Stuart–Maxwell test statistic, which requires the inversion of the nondiagonal variance–covariance matrix,  $\mathbf{V}$ . When the table is sparse, the matrix may not be of full rank, and then the command substitutes a generalized inverse  $\mathbf{V}^*$  for  $\mathbf{V}^{-1}$ . `mh` calculates optional marginal homogeneity statistics that do not require the inversion of the variance–covariance matrix. These tests may be preferred in certain situations. See [Methods and formulas](#) and [Bickeböller and Clerget-Darpoux \(1995\)](#) for details on these test statistics.

`trend` performs a test for linear trend in the (log) relative risk (RR). This option is allowed only for numeric exposure (outcome) variables, and its use should be restricted to measurements on the ordinal or the interval scales.

`cc` specifies that the continuity correction be used when calculating the test for linear trend. This correction should be specified only when the levels of the exposure variable are equally spaced.

## Remarks and examples

[stata.com](http://www.stata.com)

`symmetry` and `symmi` may be used to analyze 1-to-1 matched case-control data with multiple discrete levels of the exposure (outcome) variable.

### ► Example 1

Consider a survey of 344 individuals (BMDP 1990, 267–270) who were asked in October 1986 whether they agreed with President Reagan’s handling of foreign affairs. In January 1987, after the Iran-Contra affair became public, these same individuals were surveyed again and asked the same question. We would like to know if public opinion changed over this period.

We first describe the dataset and list a few observations.

```
. use http://www.stata-press.com/data/r15/iran
. describe
Contains data from http://www.stata-press.com/data/r15/iran.dta
  obs:      344
  vars:      2                29 Jan 2016 02:37
  size:      688
```

variable name	storage type	display format	value label	variable label
before	byte	%8.0g	vlab	Public Opinion before IC
after	byte	%8.0g	vlab	Public Opinion after IC

Sorted by:

```
. list in 1/5
```

	before	after
1.	agree	agree
2.	agree	disagree
3.	agree	unsure
4.	disagree	agree
5.	disagree	disagree

Each observation corresponds to one of the 344 individuals. The data are in wide form so that each observation has a before and an after measurement. We now perform the test without options.

#### 4 symmetry — Symmetry and marginal homogeneity tests

```
. symmetry before after
```

Public Opinion before IC	Public Opinion after IC			Total
	agree	disagree	unsure	
agree	47	56	38	141
disagree	28	61	31	120
unsure	26	47	10	83
Total	101	164	79	344

  

	chi2	df	Prob>chi2
Symmetry (asymptotic)	14.87	3	0.0019
Marginal homogeneity (Stuart-Maxwell)	14.78	2	0.0006

The test first tabulates the data in a  $K \times K$  table and then performs Bowker's (1948) test for table symmetry and the Stuart–Maxwell (Stuart 1955; Maxwell 1970) test for marginal homogeneity.

Both the symmetry test and the marginal homogeneity test are highly significant, thus indicating a shift in public opinion.

An exact test of symmetry is provided for use on sparse tables. This test is computationally intensive, so it should not be used on large tables. Because we are working on a fast computer, we will run the symmetry test again and this time include the `exact` option. We will suppress the output of the contingency table by specifying `notable` and include the `contrib` option so that we may further examine the cells responsible for the significant result.

```
. symmetry before after, contrib exact mh notable
```

Cells	Contribution to symmetry chi-squared	chi2	df	Prob>chi2
n1_2 & n2_1	9.3333			
n1_3 & n3_1	2.2500			
n2_3 & n3_2	3.2821			

  

	chi2	df	Prob>chi2
Symmetry (asymptotic)	14.87	3	0.0019
Marginal homogeneity (Stuart-Maxwell)	14.78	2	0.0006
Marginal homogeneity (Bickenboller)	13.53	2	0.0012
Marginal homogeneity (no diagonals)	15.25	2	0.0005

  

Symmetry (exact significance probability)	0.0018
---	--------

The largest contribution to the symmetry  $\chi^2$  is due to cells  $n_{12}$  and  $n_{21}$ . These correspond to changes between the agree and disagree categories. Of the 344 individuals, 56 (16.3%) changed from the agree to the disagree response, whereas only 28 (8.1%) changed in the opposite direction.

For these data, the results from the exact test are similar to those from the asymptotic test.

## ▷ Example 2

Breslow and Day (1980, 163) reprinted data from Mack et al. (1976) from a case–control study of the effect of exogenous estrogen on the risk of endometrial cancer. The data consist of 59 elderly women diagnosed with endometrial cancer and 59 disease-free control subjects living in the same community as the cases. Cases and controls were matched on age, marital status, and time living in the community. The data collected included information on the daily dose of conjugated estrogen therapy. Breslow and Day analyzed these data by creating four levels of the dose variable. Here are the data as entered into a Stata dataset:

```
. use http://www.stata-press.com/data/r15/bd163
. list, noobs divider
```

case	control	count
0	0	6
0	0.1-0.299	2
0	0.3-0.625	3
0	0.626+	1
0.1-0.299	0	9
0.1-0.299	0.1-0.299	4
0.1-0.299	0.3-0.625	2
0.1-0.299	0.626+	1
0.3-0.625	0	9
0.3-0.625	0.1-0.299	2
0.3-0.625	0.3-0.625	3
0.3-0.625	0.626+	1
0.626+	0	12
0.626+	0.1-0.299	1
0.626+	0.3-0.625	2
0.626+	0.626+	1

This dataset is in a different format from that of the [previous example](#). Instead of each observation representing one matched pair, each observation represents possibly multiple pairs indicated by the count variable. For instance, the first observation corresponds to six matched pairs where neither the case nor the control was on estrogen, the second observation corresponds to two matched pairs where the case was not on estrogen and the control was on 0.1 to 0.299 mg/day, etc.

To use `symmetry` to analyze this dataset, we must specify `fweight` to indicate that in our data there are observations corresponding to more than one matched pair.

## 6 symmetry — Symmetry and marginal homogeneity tests

```
. symmetry case control [fweight=count]
```

dosage level for case	dosage level for control				Total
	0	0.1-0.299	0.3-0.625	0.626+	
0	6	2	3	1	12
0.1-0.299	9	4	2	1	16
0.3-0.625	9	2	3	1	15
0.626+	12	1	2	1	16
Total	36	9	10	4	59

  

	chi2	df	Prob>chi2
Symmetry (asymptotic)	17.10	6	0.0089
Marginal homogeneity (Stuart-Maxwell)	16.96	3	0.0007

Both the test of symmetry and the test of marginal homogeneity are highly significant, thus leading us to reject the null hypothesis that there is no effect of exposure to estrogen on the risk of endometrial cancer.

Breslow and Day perform a test for trend assuming that the estrogen exposure levels were equally spaced by recoding the exposure levels as 1, 2, 3, and 4.

We can easily reproduce their results by recoding our data in this way and by specifying the trend option. Two new numeric variables were created, ca and co, corresponding to the variables case and control, respectively. Below we list some of the data and our results from symmetry:

```
. encode case, gen(ca)
. encode control, gen(co)
. label values ca
. label values co
. list in 1/4
```

	case	control	count	ca	co
1.	0	0	6	1	1
2.	0	0.1-0.299	2	1	2
3.	0	0.3-0.625	3	1	3
4.	0	0.626+	1	1	4

```
. symmetry ca co [fw=count], notable trend cc
```

	chi2	df	Prob>chi2
Symmetry (asymptotic)	17.10	6	0.0089
Marginal homogeneity (Stuart-Maxwell)	16.96	3	0.0007
Linear trend in the (log) RR	14.43	1	0.0001

We requested the continuity correction by specifying cc. Doing so is appropriate because our coded exposure levels are equally spaced.

The test for trend was highly significant, indicating an increased risk of endometrial cancer with increased dosage of conjugated estrogen.

You must be cautious: the way in which you code the exposure variable affects the linear trend statistic. If instead of coding the levels as 1, 2, 3, and 4, we had instead used 0, 0.2, 0.46, and 0.7

(roughly the midpoint in the range of each level), we would have obtained a  $\chi^2$  statistic of 11.19 for these data.



## Stored results

`symmetry` stores the following in `r()`:

Scalars

<code>r(N_pair)</code>	number of matched pairs
<code>r(chi2)</code>	asymptotic symmetry $\chi^2$
<code>r(df)</code>	asymptotic symmetry degrees of freedom
<code>r(p)</code>	asymptotic symmetry $p$ -value
<code>r(chi2_sm)</code>	MH (Stuart–Maxwell) $\chi^2$
<code>r(df_sm)</code>	MH (Stuart–Maxwell) degrees of freedom
<code>r(p_sm)</code>	MH (Stuart–Maxwell) $p$ -value
<code>r(chi2_b)</code>	MH (Bickenböller) $\chi^2$
<code>r(df_b)</code>	MH (Bickenböller) degrees of freedom
<code>r(p_b)</code>	MH (Bickenböller) $p$ -value
<code>r(chi2_nd)</code>	MH (no diagonals) $\chi^2$
<code>r(df_nd)</code>	MH (no diagonals) degrees of freedom
<code>r(p_nd)</code>	MH (no diagonals) $p$ -value
<code>r(chi2_t)</code>	$\chi^2$ for linear trend
<code>r(p_trend)</code>	$p$ -value for linear trend
<code>r(p_exact)</code>	exact symmetry $p$ -value

## Methods and formulas

Methods and formulas are presented under the following headings:

*Asymptotic tests*

*Exact symmetry test*

### Asymptotic tests

Consider a square table with  $K$  exposure categories, that is,  $K$  rows and  $K$  columns. Let  $n_{ij}$  be the count corresponding to row  $i$  and column  $j$  of the table,  $N_{ij} = n_{ij} + n_{ji}$ , for  $i, j = 1, 2, \dots, K$ , and  $n_{i.}$ , and let  $n_{.j}$  be the marginal totals for row  $i$  and column  $j$ , respectively. Asymptotic tests for symmetry and marginal homogeneity for this  $K \times K$  table are calculated as follows:

The null hypothesis of complete symmetry  $p_{ij} = p_{ji}$ ,  $i \neq j$ , is tested by calculating the test statistic (Bowker 1948)

$$T_{cs} = \sum_{i < j} \frac{(n_{ij} - n_{ji})^2}{n_{ij} + n_{ji}}$$

which is asymptotically distributed as  $\chi^2$  with  $K(K - 1)/2 - R$  degrees of freedom, where  $R$  is the number of off-diagonal cells with  $N_{ij} = 0$ .

The null hypothesis of marginal homogeneity,  $p_{i.} = p_{.i}$ , is tested by calculating the Stuart–Maxwell test statistic (Stuart 1955; Maxwell 1970),

$$T_{sm} = \mathbf{d}'\mathbf{V}^{-1}\mathbf{d}$$

where  $\mathbf{d}$  is a column vector with elements equal to the differences  $d_i = n_{i.} - n_{.i}$  for  $i = 1, 2, \dots, K$ , and  $\mathbf{V}$  is the variance–covariance matrix with elements

$$\begin{aligned} v_{ii} &= n_{i.} + n_{.i} - 2n_{ii} \\ v_{ij} &= -(n_{ij} + n_{ji}), \quad i \neq j \end{aligned}$$

$T_{sm}$  is asymptotically  $\chi^2$  with  $K - 1$  degrees of freedom.

This test statistic properly accounts for the dependence between the table's rows and columns. When the matrix  $\mathbf{V}$  is not of full rank, a generalized inverse  $\mathbf{V}^*$  is substituted for  $\mathbf{V}^{-1}$ .

The Bickeböllner and Clerget-Darpoux (1995) marginal homogeneity test statistic is calculated by

$$T_{mh} = \sum_i \frac{(n_{i.} - n_{.i})^2}{n_{i.} + n_{.i}}$$

This statistic is asymptotically distributed, under the assumption of marginal independence, as  $\chi^2$  with  $K - 1$  degrees of freedom.

The marginal homogeneity (no diagonals) test statistic  $T_{mh}^0$  is calculated in the same way as  $T_{mh}$ , except that the diagonal elements do not enter into the calculation of the marginal totals. Unlike the previous test statistic,  $T_{mh}^0$  reduces to a McNemar test statistic for  $2 \times 2$  tables. The test statistic  $\{(K - 1)/2\}T_{mh}^0$  is asymptotically distributed as  $\chi^2$  with  $K - 1$  degrees of freedom (Cleves, Olson, and Jacobs 1997; Spielman and Ewens 1996).

Breslow and Day's test statistic for linear trend in the (log) of RR is

$$\frac{\left\{ \sum_{i < j} (n_{ij} - n_{ji})(X_j - X_i) - cc \right\}^2}{\sum_{i < j} (n_{ij} + n_{ji})(X_j - X_i)^2}$$

where the  $X_j$  are the doses associated with the various levels of exposure and  $cc$  is the continuity correction; it is asymptotically distributed as  $\chi^2$  with 1 degree of freedom.

The continuity correction option is applicable only when the levels of the exposure variable are equally spaced.

## Exact symmetry test

The exact test is based on a permutation algorithm applied to the null distribution. The distribution of the off-diagonal elements  $n_{ij}$ ,  $i \neq j$ , conditional on the sum of the complementary off-diagonal cells,  $N_{ij} = n_{ij} + n_{ji}$ , can be written as the product of  $K(K - 1)/2$  binomial random variables,

$$P(\mathbf{n}) = \prod_{i < j} \binom{N_{ij}}{n_{ij}} \pi_{ij}^{n_{ij}} (1 - \pi_{ij})^{n_{ji}}$$

where  $\mathbf{n}$  is a vector with elements  $n_{ij}$  and  $\pi_{ij} = E(n_{ij}/N_{ij}|N_{ij})$ . Under the null hypothesis of complete symmetry,  $\pi_{ij} = \pi_{ji} = 1/2$ , and thus the permutation distribution is given by



$$P_0(\mathbf{n}) = \prod_{i < j} \binom{N_{ij}}{n_{ij}} \left(\frac{1}{2}\right)^{N_{ij}}$$

The exact significance test is performed by evaluating

$$P_{cs} = \sum_{\mathbf{n} \in p} P_0(\mathbf{n})$$

where  $p = \{\mathbf{n} : P_0(\mathbf{n}) < P_0(\mathbf{n}^*)\}$  and  $\mathbf{n}^*$  is the observed contingency table data vector. The algorithm evaluates  $p_{cs}$  exactly. For information about permutation tests, see Good (2005, 2006).

## References

- Bickeböller, H., and F. Clerget-Darpoux. 1995. Statistical properties of the allelic and genotypic transmission/disequilibrium test for multiallelic markers. *Genetic Epidemiology* 12: 865–870.
- BMDP. 1990. *BMDP Statistical Software Manual*. Oakland, CA: University of California Press.
- Bowker, A. H. 1948. A test for symmetry in contingency tables. *Journal of the American Statistical Association* 43: 572–574.
- Breslow, N. E., and N. E. Day. 1980. *Statistical Methods in Cancer Research: Vol. 1—The Analysis of Case-Control Studies*. Lyon: IARC.
- Cleves, M. A. 1997. [sg74: Symmetry and marginal homogeneity test/Transmission-Disequilibrium Test \(TDT\)](#). *Stata Technical Bulletin* 40: 23–27. Reprinted in *Stata Technical Bulletin Reprints*, vol. 7, pp. 193–197. College Station, TX: Stata Press.
- . 1999. [sg110: Hardy–Weinberg equilibrium test and allele frequency estimation](#). *Stata Technical Bulletin* 48: 34–37. Reprinted in *Stata Technical Bulletin Reprints*, vol. 8, pp. 280–284. College Station, TX: Stata Press.
- Cleves, M. A., J. M. Olson, and K. B. Jacobs. 1997. Exact transmission-disequilibrium tests with multiallelic markers. *Genetic Epidemiology* 14: 337–347.
- Cui, J. 2000. [sg150: Hardy–Weinberg equilibrium test in case–control studies](#). *Stata Technical Bulletin* 57: 17–19. Reprinted in *Stata Technical Bulletin Reprints*, vol. 10, pp. 218–220. College Station, TX: Stata Press.
- Good, P. I. 2005. *Permutation, Parametric, and Bootstrap Tests of Hypotheses: A Practical Guide to Resampling Methods for Testing Hypotheses*. 3rd ed. New York: Springer.
- . 2006. *Resampling Methods: A Practical Guide to Data Analysis*. 3rd ed. Boston: Birkhäuser.
- Mack, T. M., M. C. Pike, B. E. Henderson, R. I. Pfeffer, V. R. Gerkins, M. Arthur, and S. E. Brown. 1976. Estrogens and endometrial cancer in a retirement community. *New England Journal of Medicine* 294: 1262–1267.
- Mander, A. P. 2000. [sbe38: Haplotype frequency estimation using an EM algorithm and log-linear modeling](#). *Stata Technical Bulletin* 57: 5–7. Reprinted in *Stata Technical Bulletin Reprints*, vol. 10, pp. 104–107. College Station, TX: Stata Press.
- Maxwell, A. E. 1970. Comparing the classification of subjects by two independent judges. *British Journal of Psychiatry* 116: 651–655.
- Spieldman, R. S., and W. J. Ewens. 1996. The TDT and other family-based tests for linkage disequilibrium and association. *American Journal of Human Genetics* 59: 983–989.
- Spieldman, R. S., R. E. McGinnis, and W. J. Ewens. 1993. Transmission test for linkage disequilibrium: The insulin gene region and insulin-dependence diabetes mellitus (IDDM). *American Journal of Human Genetics* 52: 506–516.
- Stuart, A. 1955. A test for homogeneity of the marginal distributions in a two-way classification. *Biometrika* 42: 412–416.

## Also see

[R] [epitab](#) — Tables for epidemiologists